# Solutions to some exercises

## Stable, center, unstable subspaces

### 1 (a)

In[ ]:=
```
A = {{2, 3}, {0, -1}};
Eigenvalues[A]
Eigenvectors[A]
StreamPlot[A.{x, y}, {x, -2, 2}, {y, -2, 2}]
```

Out[ ]=
```
{2, -1}
```

Out[ ]=
```
{{1, 0}, {-1, 1}}
```

Out[ ]=



# 1 (b)

In[ ]:=
```
A = {{2, 4}, {0, -2}};
Eigenvalues[A]
Eigenvectors[A]
StreamPlot[A.{x, y}, {x, -2, 2}, {y, -2, 2}]
```

Out[ ]=
```
{-2, 2}
```

Out[ ]=
```
{{-1, 1}, {1, 0}}
```

Out[ ]=




# 1 (c)

In[ ]:=
```
A = {{-1, -3, 0}, {0, 2, 0}, {0, 0, -1}};
Eigenvalues[A]
Eigenvectors[A]
```

Out[ ]=
```
{2, -1, -1}
```

Out[ ]=
```
{{-1, 1, 0}, {0, 0, 1}, {1, 0, 0}}
```

## 1 (d)

In[ ]:=

```
A = {{2, 3, 0}, {0, -1, 0}, {0, 0, -1}};
Eigenvalues[A]
Eigenvectors[A]
```

Out[ ]=

```
{2, -1, -1}
```

Out[ ]=

```
{{1, 0, 0}, {0, 0, 1}, {-1, 1, 0}}
```

## 3 (a)

False. The solution starting at the origin also belongs to $E^u$. But the bigger problem with the statement is that there could be solutions starting outside of $E^u$, and still the distance from the origin is going to infinity.

## 3 (b)

True.

## 3 (c)

False. The solution starting at the origin is an exception.

## 3 (d)

False. Not true for ellipses. Also, problematic when 0 is an eigenvalue.

## 3 (e)

False. Not true e.g. when there is a $2m \times 2m$ Jordan block with $m \geq 2$ corresponding to a pair of purely imaginary eigenvalues, but also problematic when 0 is an eigenvalue.

## 3 (f)

True.

## 3 (g)

False. The forward trajectories need not be bounded for solutions in $E^c$.

## 3 (h)

False. The trajectories need not be bounded for solutions in $E^c$.

## 3 (i)

False. There could be all kind of solutions outside $E^c$ that do not converge to the origin as $t \to \infty$ or $t \to -\infty$.

## Non-hyperbolic linear systems

### 4 (a)

Three-dimensional. These are matrices with $\det A > 0$ and $\operatorname{tr} A = 0$. The general form is $\{\{a, b\}, \{c, -a\}\}$ with $-a^2 - b\,c > 0$.

In[ ]:=
```
B = {{0, -1}, {1, 0}};
P = {{1, 2}, {0, 1}};
A = P.B.Inverse[P];
MatrixForm[A]
line = Plot[x Tan[π/8], {x, -2, 2}, PlotStyle → Magenta];
strpl = StreamPlot[A.{x, y}, {x, -2, 2}, {y, -2, 2}];
Show[strpl, line]
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 2 & -5 \\ 1 & -2 \end{pmatrix}$$

Out[ ]=




Remark: The general form of an ellipse is $A\,x^2 + B\,x\,y + C\,y^2 = R$ with $A > 0$, $C > 0$, $A\,C > \frac{B^2}{4}$, $R > 0$.

If $A = C$ and $B = 0$ then it is a circle. If $A = C$ and $B \neq 0$ then it is an ellipse whose axes have slopes $\pm 45^\circ$.

Otherwise, rotation by $\varphi = \frac{1}{2} \operatorname{arctg} \frac{B}{C - A} \in \left(-\frac{\pi}{4}, \frac{\pi}{4}\right)$ brings the ellipse to one with horizontal/vertical axes.

E.g. for the above example one finds that $B = -4A$ and $C = 5A$ (in the formula of the ellipse), and thus $\varphi = -\frac{\pi}{8}$.

How can one find $B = -4A$ and $C = 5A$? Differentiate both sides of $A x^2 + B x y + C y^2 = R$ w.r.t. time to get $2 A \dot{x} x + B(\dot{x} y + x \dot{y}) + 2 C \dot{y} y = 0$, and plug in for $\dot{x}$ and $\dot{y}$ the ODE.

## 4 (b)

Two-dimensional (but it is not really a manifold). These are matrices with $\det A = 0$ and $\operatorname{tr} A = 0$.

## 4 (c)

Two-dimensional. These are matrices with $\det A = 0$, $\operatorname{tr} A = 0$, and $A \neq 0$. This set is the union of the two two-dimensional manifolds
$\{\{-a\,c, a^2\}, \{-c^2, a\,c\}\}$ and $\{\{a\,c, -a^2\}, \{c^2, -a\,c\}\}$ (in both cases, at least one of $a$ and $c$ is positive).

In[ ]:=
```
B = {{0, 1}, {0, 0}};
P = {{-1, 1}, {2, 1}};
A = P.B.Inverse[P];
MatrixForm[A]
line = Plot[-2 x , {x, -2, 2}, PlotStyle → Magenta];
strpl = StreamPlot[A.{x, y}, {x, -2, 2}, {y, -2, 2}];
Show[strpl, line]
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} -\frac{2}{3} & -\frac{1}{3} \\ \frac{4}{3} & \frac{2}{3} \end{pmatrix}$$

Out[ ]=



### 4 (d)

Zero-dimensional. Only the zero matrix. Every point in the plane is an equilibrium.

### 5

$\det A > 0$ and $\operatorname{tr} A = 0$, see exercise 4 (a)

### 6

In[ ]:=
```
Expand[(x - ρ) (x - ⅈ ω) (x + ⅈ ω)]
```

Out[ ]=

$x^3 - x^2 \rho + x \omega^2 - \rho \omega^2$

Based on this, one finds the condition $b_2 > 0, b_0 > 0, b_1 b_2 = b_0$. Since $b_2 = -\operatorname{tr} A, b_0 = -\det A, b_1 = M$, we find the equivalent condition is $\det A < 0, \operatorname{tr} A < 0, \det A = M \operatorname{tr} A$.

## Routh-Hurwitz criterion

### 7

In[ ]:=
```
H = {{b3, 1, 0, 0}, {b1, b2, b3, 1}, {0, b0, b1, b2}, {0, 0, 0, b0}};
h1 = H[[1, 1]]
h2 = Det[H[[{1, 2}, {1, 2}]]]
h3 = Det[H[[{1, 2, 3}, {1, 2, 3}]]]
h4 = Simplify[Det[H]]
Reduce[{h1, h2, h3, h4} > 0]
```

Out[ ]=

$b_3$

Out[ ]=

$-b_1 + b_2 b_3$

Out[ ]=

$-b_1^2 + b_1 b_2 b_3 - b_0 b_3^2$

Out[ ]=

$-b_0 \left(b_1^2 - b_1 b_2 b_3 + b_0 b_3^2\right)$

Out[ ]=

$b_1 > 0 \;\&\&\; b_3 > 0 \;\&\&\; b_0 > 0 \;\&\&\; b_2 > \dfrac{b_1^2 + b_0 b_3^2}{b_1 b_3}$

## Quadratic Lyapunov function for linear systems with a stable matrix

### 8

In[ ]:=
```
A = {{-1, 1}, {0, -1}};
Q = Integrate[MatrixExp[(A + Aᵀ) t], {t, 0, ∞}];
MatrixForm[Q]
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} \frac{2}{3} & \frac{1}{3} \\ \frac{1}{3} & \frac{2}{3} \end{pmatrix}$$

Alternative solution: first find the Jordan decomposition of $A + A^T$ and then compute $e^{(A+A^T)t} = P\, e^{B\,t}\, P^{-1}$, where $B = P^{-1}\,(A + A^T)\, P$.

In[ ]:=
```
A = {{-1, 1}, {0, -1}};
{P, B} = JordanDecomposition[A + Aᵀ];
Print["P =", MatrixForm[P], " and B =", MatrixForm[B]];
P.B.Inverse[P] == A + Aᵀ (* verify *)
Q = Integrate[P.MatrixExp[B t].Inverse[P], {t, 0, ∞}];
MatrixForm[Q]
```

$P = \begin{pmatrix} -1 & 1 \\ 1 & 1 \end{pmatrix}$ and $B = \begin{pmatrix} -3 & 0 \\ 0 & -1 \end{pmatrix}$

Out[ ]=

```
True
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} \frac{2}{3} & \frac{1}{3} \\ \frac{1}{3} & \frac{2}{3} \end{pmatrix}$$

Yet another solution (by the Newton-Leibniz rule):

In[ ]:=
```
A = {{-1, 1}, {0, -1}};
MatrixForm[-Inverse[A + Aᵀ]]
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} \frac{2}{3} & \frac{1}{3} \\ \frac{1}{3} & \frac{2}{3} \end{pmatrix}$$

### 9

Let $Q$ be the identity (with this, $Q\,A + A^T\,Q$ is negative definite). Then $S$ is the standard unit sphere. Further, since the eigenvalues of $A$ are $-1 \pm i$ (and $A$ is in Jordan canonical form), the solution starting from $p$ is $e^{-t} R(t) p$, where $R(t)$ is a rotation matrix (rotation by $t$ radian). Hence, the length of the solution is $e^{-t} \| p \|$, and thus, $\tau(p) = \log \| p \|$. Further, $h(p) = R(\log \| p \|) p$ and $h^{-1}(q) = R(-\log \| q \|) q$.

## An equilibrium that is attracting, but not Lyapunov stable

### 11

In[∘]:=
```
g = {r (1 - r), r (1 - Cos[ϕ])};
f = Simplify[{x g[[1]]/r - y g[[2]], y g[[1]]/r + x g[[2]]} /. {Cos[ϕ] → x/r}]
StreamPlot[f /. {r → Sqrt[x^2 + y^2]}, {x, -2, 2}, {y, -2, 2}]
```

Out[∘]=

$\{-r\,y + x\,(1 - r + y),\ -x^2 + r\,(x - y) + y\}$

Out[∘]=



# Lyapunov function

## 12

In[ ]:=
```
f = {-2 y + y z - x^3, x - x z - y^3, x y - z^3};
V = a x^2 + b y^2 + c z^2;
Vdot = Expand[D[V, {{x, y, z}}].f]
monomials = MonomialList[Vdot, {x, y, z}]
```

Out[ ]=

$$-2\,a\,x^4 - 4\,a\,x\,y + 2\,b\,x\,y - 2\,b\,y^4 + 2\,a\,x\,y\,z - 2\,b\,x\,y\,z + 2\,c\,x\,y\,z - 2\,c\,z^4$$

Out[ ]=

$$\left\{-2\,a\,x^4,\ (2\,a - 2\,b + 2\,c)\,x\,y\,z,\ (-4\,a + 2\,b)\,x\,y,\ -2\,b\,y^4,\ -2\,c\,z^4\right\}$$

In[ ]:=
```
d1 = Coefficient[monomials[[2]], x y z];
d2 = Coefficient[monomials[[3]], x y];
FindInstance[d1 == 0 && d2 == 0 && {a, b, c} > 0, {a, b, c}]
```

Out[ ]=

$$\{\{a \to 1,\ b \to 2,\ c \to 1\}\}$$

## 13

In[ ]:=
```
f = {-y - x y^2 + z^2 - x^3, x + z^3 - y^3, -x z - z x^2 - y z^2 - z^5};
V = x^2 + y^2 + z^2;
Simplify[D[V, {{x, y, z}}].f]
```

Out[ ]=

$$-2\left(x^4 + y^4 + z^6 + x^2\left(y^2 + z^2\right)\right)$$

In[ ]:=
```
A = D[f, {{x, y, z}}] /. {x → 0, y → 0, z → 0};
MatrixForm[A]
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 0 & -1 & 0 \\ 1 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

## 14 (a)

In[ ]:=
```
f = {-x + y + x y, x - y - x^2 - y^3};
Reduce[f == 0, {x, y} ∈ Reals]
V = x^2 + y^2;
Simplify[D[V, {{x, y}}].f]
```

Out[ ]=

$x == 0 \,\&\&\, y == 0$

Out[ ]=

$-2\left(x^2 - 2\,x\,y + y^2 + y^4\right)$

In[ ]:=
```
StreamPlot[f, {x, -2, 2}, {y, -2, 2}]
```

Out[ ]=



## 14 (b)

In[ ]:=
```
f = {-x - 2 y + x y^2, 3 x - 3 y + y^3};
Reduce[f == 0, {x, y} ∈ Reals]
V = x^2 + b y^2;
Vdot = D[V, {{x, y}}].f;
MonomialList[Vdot, {x, y}]
```

Out[ ]=

$x == 0 \,\&\&\, y == 0$

Out[ ]=

$\{2 x^2 y^2, -2 x^2, (-4 + 6 b) x y, 2 b y^4, -6 b y^2\}$

In[ ]:=
```
MonomialList[Vdot /. {b -> 2/3}]
Reduce[(Vdot /. {b -> 2/3}) > 0 && x^2 + y^2 < 1]
Plot3D[Vdot /. {b -> 2/3}, {x, y} ∈ Disk[{0, 0}, 2]]
```

Out[ ]=

$\left\{2 x^2 y^2, -2 x^2, \frac{4 y^4}{3}, -4 y^2\right\}$

Out[ ]=

False

Out[ ]=



In[ ]:=
```
StreamPlot[f, {x, -5, 5}, {y, -5, 5}]
```

Out[ ]=




## 15

In[ ]:=
```
f = {y, -q[x]};
```

$$V = \frac{y^2}{2} + \text{Integrate}[q[\alpha], \{\alpha, 0, x\}];$$

```
D[V, {{x, y}}].f
```

Out[ ]=

```
0
```

# Sink, source, saddle

## 16 (a)

In[ ]:=
```
f = {x - x y, y - x^2};
equil = Solve[f == 0]
A = D[f, {{x, y}}];
Eigenvalues[A /. equil[[1]]]
Eigenvalues[A /. equil[[2]]]
Eigenvalues[A /. equil[[3]]]
StreamPlot[f, {x, -2, 2}, {y, -2, 2}, GridLines → Automatic]
```

Out[ ]=
```
{{x → -1, y → 1}, {x → 0, y → 0}, {x → 1, y → 1}}
```

Out[ ]=
```
{2, -1}
```

Out[ ]=
```
{1, 1}
```

Out[ ]=
```
{2, -1}
```

Out[ ]=



## 16 (b)

In[ ]:=

```
f = {-4 y + 2 x y - 8, 4 y^2 - x^2};
equil = Solve[f == 0, {x, y} ∈ Reals]
A = D[f, {{x, y}}];
Eigenvalues[A /. equil[[1]]]
Eigenvalues[A /. equil[[2]]]
StreamPlot[f, {x, -5, 5}, {y, -5, 5}, GridLines → Automatic]
```

Out[ ]=

$\{\{x \to -2, y \to -1\}, \{x \to 4, y \to 2\}\}$

Out[ ]=

$\{-5 + i\sqrt{23}, -5 - i\sqrt{23}\}$

Out[ ]=

$\{12, 8\}$

Out[ ]=



## 16 (c)

In[ ]:=

```
f = {2 x - 2 x y, 2 y - x^2 + y^2};
equil = Solve[f == 0, {x, y} ∈ Reals]
A = D[f, {{x, y}}];
Eigenvalues[A /. equil[[1]]]
Eigenvalues[A /. equil[[2]]]
Eigenvalues[A /. equil[[3]]]
Eigenvalues[A /. equil[[4]]]
StreamPlot[f, {x, -3, 3}, {y, -3, 3}, GridLines → Automatic]
```

Out[ ]=

$\{\{x \to 0, y \to -2\}, \{x \to 0, y \to 0\}, \{x \to -\sqrt{3}, y \to 1\}, \{x \to \sqrt{3}, y \to 1\}\}$

Out[ ]=

$\{6, -2\}$

Out[ ]=

$\{2, 2\}$

Out[ ]=

$\{6, -2\}$

Out[ ]=

$\{6, -2\}$

Out[ ]=



## 16 (d)

In[ ]:=
```
f = {-x, -y + x^2, z + x^2};
equil = Solve[f == 0, {x, y, z} ∈ Reals]
A = D[f, {{x, y, z}}];
Eigenvalues[A /. equil[[1]]]
```

Out[ ]=
```
{{x → 0, y → 0, z → 0}}
```

Out[ ]=
```
{-1, -1, 1}
```

## 16 (e)

In[ ]:=
```
f = {y - x, α x - y - x z, x y - z};
A = D[f, {{x, y, z}}];
equil = Solve[f == 0 && α < 1, {x, y, z} ∈ Reals]
Eigenvalues[A /. Normal[equil[[1]]]]
```

Out[ ]=
```
{{x → 0 if α < 1, y → 0 if α < 1, z → 0 if α < 1}}
```

Out[ ]=

$\{-1, -1 - \sqrt{\alpha}, -1 + \sqrt{\alpha}\}$

In[ ]:=
```
equil = Solve[(f /. {α → 1}) == 0, {x, y, z}]
Eigenvalues[A /. {α → 1} /. equil[[1]]]
```

Out[ ]=
```
{{x → 0, y → 0, z → 0}}
```

Out[ ]=
```
{-2, -1, 0}
```

In[ ]:=
```
equil = Solve[f == 0 && α > 1, {x, y, z} ∈ Reals]
Eigenvalues[A /. Normal[equil[[1]]]]
Eigenvalues[A /. Normal[equil[[2]]]]
Eigenvalues[A /. Normal[equil[[3]]]]
```

Out[ ]=

$$\{\{x \to 0 \text{ if } \alpha > 1,\ y \to 0 \text{ if } \alpha > 1,\ z \to 0 \text{ if } \alpha > 1\},$$
$$\{x \to -\sqrt{-1+\alpha} \text{ if } \alpha > 1,\ y \to -\sqrt{-1+\alpha} \text{ if } \alpha > 1,\ z \to -1+\alpha \text{ if } \alpha > 1\},$$
$$\{x \to \sqrt{-1+\alpha} \text{ if } \alpha > 1,\ y \to \sqrt{-1+\alpha} \text{ if } \alpha > 1,\ z \to -1+\alpha \text{ if } \alpha > 1\}\}$$

Out[ ]=

$$\{-1,\ -1-\sqrt{\alpha},\ -1+\sqrt{\alpha}\}$$

Out[ ]=

$$\left\{-2,\ \frac{1}{2}\left(-1-\sqrt{5-4\alpha}\right),\ \frac{1}{2}\left(-1+\sqrt{5-4\alpha}\right)\right\}$$

Out[ ]=

$$\left\{-2,\ \frac{1}{2}\left(-1-\sqrt{5-4\alpha}\right),\ \frac{1}{2}\left(-1+\sqrt{5-4\alpha}\right)\right\}$$